
EQSUB (NAME=アウトプットの方程式名, PRINT)
 インプットの方程式名 マクロ方程式名のリスト ;

機能 :

EQSUBは1つ以上の方程式を他の方程式に置き換えます。このコマンドはパラメータに制約がある場合の推定や、共通の項を持つ長い式の推定や、あるいは後にインプット方程式の1つが変化するだけで外生変数が変化できるような複雑なモデルを組み立てるのに有用です。

使用法:

マクロ方程式(主方程式)とインプット方程式をFRML, あるいはIDENTステートメントで定義します。EQSUBは、各インプット方程式をマクロ方程式に、左から右へ代入します。各インプット方程式に対して、EQSUBはマクロ方程式の引数リストにある従属変数(あるいは従属変数がなければ、方程式名)を探します。従属変数が見つければ、インプット方程式のコードがマクロ方程式に挿入され、旧変数名は削除されます。たとえば

```
FRML EQ1 Y=A+XB;
FRML EXB XB=X1*B1 + X2*B2;
EQSUB EQ1 EXB;
```

は、次の式と同じになります。

```
FRML EQ1 Y=A+X1*B1 + X2*B2;
```

結果として得られた方程式は、NAME=オプションが与えられていない限りマクロ方程式と入れ替えられます。いくつかの異なるマクロ方程式がある場合は、DOTコマンドが役に立ちます。

多くの要素があるインプット方程式を定義するのに、従属変数名を残しておくと便利です。そうすれば、従属変数とそれぞれの方程式名の両方を作成する必要はありません。そのような方程式を、TSPでは非正規化されていると呼びます。たとえば

```
FRML E Y1-XB; FRML XB B0+B1*X1 + B2*X2; ?change XB to add/delete exog. variables
FRML TOBIT LOGL=YPOS*(LNORM(E/SIGMA)-LOG(SIGMA))+YZERO*LCNORM(-XB/SIGMA);
EQSUB(NAME=TOBIT1) TOBIT E XB;
```

TOBITとEは両方ともXBに依存しますので、XBは最後に代入されることに注意して下さい。XBをEに個別に代入する必要はありません。個別に代入してももちろん正しく作動しますが、結果はより大きく、非効率なコードになる場合もあります。新しいFRML TOBIT1が作成され、もともとのTOBITは後で使用するために手を付けずに残されます。

他方、正規化されたインプット方程式はパラメータ制約のためには推奨できます。同じFRMLが通常パラメータ制約を課すためと、推定した後でANALYZで制約を課したパラメータを評価するための両方に用いられます。もしインプット方程式が正規化されていれば、ANALYZは制約を課したパラメータ名を保存することができます。たとえば、対称性の制約をおいたトランスログモデルでは、次のようになります。

EQSUB

```
FRML EQ1 SH1=A1+B11*LP1+B12*LP2+B13*LP3;
FRML EQ2 SH2=A2+B21*LP1+B22*LP2+B23*LP3;
?note: The last share equation is not used in estimation due to singularity.
?FRML EQ3 SH3=A3+B31*LP1+B32*LP2+B33*LP3;
FRML R13 B13=-(B11+B12);      ? homogeneity/adding up constraints
FRML R23 B23=-(B12+B22);
EQSUB EQ1 R13; EQSUB EQ2 R23;
FRML L01 A3=1-(A1+A2);      ? left out params from final equation
FRML L02 B33=-(B12+B23);    ? note: this also depends on the restricted B23
EQSUB L02 R23;
LSQ EQ1 EQ2;                ? Estimate model with restrictions in place
ANALYZ L01 L02;             ? Print and store values and standard errors for A3 and B33
```

EQSUBは、「ラグ付従属変数」も扱うことができます(1996/4から).

```
FRML U Y-(A+B*X+G*Z(-2));
FRML E U-RHO*U(-1);
EQSUB E U;
```

この式は、次式と同じです.

```
FRML E Y-(A+B*X+G*Z(-2))-RHO*(Y(-1)-(A+B*X(-1)+G*Z(-3)));
```

EQSUBコマンドが与えられたときには、方程式にあるすべての変数が定義されていることが必要です(系列, PARAM, CONSTとしてか、あるいは他のFRMLとして)、従ってEQSUBはラグをとる必要がある(系列とFRML)ものと、ラグをとる必要がない(PARAMとCONST)ものが分かっています.

オプション:

NAME=新アウトプット方程式名 は、アウトプット方程式に新しい名前を与えます. このオプションがなければ、インプット方程式が置き換えられます.

PRINT/NOPRINT アウトプット方程式をプリントするかどうか指定します.

例:

上記参照. また、MLで推定するために対数尤度関数をセットアップするのにEQSUBを用いるいろいろな例についてはTSP ユーザース・ガイドも参照つぎは、もう一つの例でインプット方程式F1からF20を主方程式E1からE8へ代入するのにDOTを用いる方法を示したものです.

```
DOT E1-E8;
EQSUB . F1-F20;
ENDDOT;
```

アウトプット:

通常アウトプット方程式は表示なしに保存され、インプット方程式を置き換えるか、新しい方程式を作成します. もしPRINTオプションが指定されていると、アウトプット方程式がプリントされます.